

Three Shorts – Fini!

Once again, kick off your boots and soothe that aching head to the sights and sounds of ON THREE! The first two programs are graphic demos, while the last is a sound demo! It will give you a chorus of weird noises.

To use the program "Bob's Blocks", just type it and enter "RUN". No other files are needed. For the program "Bob's Lines", you need to have the file "BGRAF.INV" on a disk named "/BASIC" in one of your drives. If this file is elsewhere, make the appropriate change to line #100. If the program can not find that file, it will hang, and you will have to press the RESET key to stop it.

The program "Bob's Noises??" uses the ".AUDIO" driver to generate weird sounding noises. If the ".AUDIO" driver is not configured into your system, you will get an error message.

ON THREE will pay \$25 for any short demonstration program used in this space, so send in your favorite today, and we will see you next time in ON THREE. *///*

```

0 REM #####
1 REM # Bob's Blocks #
2 REM # ----- #
3 REM # This simple little program shows how #
4 REM # fast the Apple /// can display diff- #
5 REM # erent colors on the screen. Just #
6 REM # type in the program and enter 'RUN'. #
7 REM # This works best with a color monitor #
8 REM # but is still okay without one. #
9 REM #####
10 x.left%=0:x.right%=0:y.top%=0:y.bot%=0
20 loop.color%=0:back.color%=20:clear.view%=28
30 black%=0:white%=15
40 text.mode%=16:color.mode%=1
50 PRINT CHR$(text.mode%);CHR$(color.mode%)
60 HOME
99 ON KBD GOTO 1000
100 FOR loop.color%=black% TO white%
110 GOSUB 200
120 PRINT CHR$(back.color%);CHR$(loop.color%);
130 PRINT CHR$(clear.view%);
140 NEXT loop.color%
150 GOTO 100
200 x.left%=RND(1)*40:x.right%=RND(1)*40
210 IF x.left%>x.right% THEN SWAP x.left%,x.right%
220 y.top%=RND(1)*24:y.bot%=RND(1)*24
230 IF y.top%>y.bot% THEN SWAP y.top%,y.bot%
240 WINDOW x.left%,y.top% TO x.right%,y.bot%
250 RETURN
1000 IF KBD=27 THEN POP:TEXT:HOME:END
1010 ON KBD GOTO 1000
1020 RETURN
    
```

```

0 REM #####
1 REM # Bob's Lines #
2 REM # ----- #
3 REM # Another very short program, it will #
4 REM # draw lines on your graphics screen. #
5 REM # It pickes random endpoints for the #
6 REM # lines so you will get a colorful #
7 REM # straw affect. To use, make sure you #
8 REM # have '/BASIC/BGRAF.INV' on line. #
9 REM #####
100 ON ERR INVOKE"/BASIC/BGRAF.INV"
110 PERFORM initgrafix:OFF ERR
120 xdist%=140:ydist%=192:mode%=3:buf%=1
130 PERFORM grafixmode(%mode%,%buf%):PERFORM grafixon
140 PERFORM fillport
199 ON KBD GOTO 1000
200 FOR a%=0 TO 1000
210 PERFORM pencolor(%RND(1)*16)
220 PERFORM moveto(%RND(1)*xdist%,%RND(1)*ydist%)
230 PERFORM lineto(%RND(1)*xdist%,%RND(1)*ydist%)
240 NEXT
250 GOTO 200
1000 IF KBD=27 THEN POP:TEXT:HOME:END
1010 ON KBD GOTO 1000
1020 RETURN
    
```

```

0 REM #####
1 REM # Bob's Noises?? #
2 REM # ----- #
3 REM # Up to now we have used this column to #
4 REM # show off the graphics capabilities of #
5 REM # the ///. This program shows you some #
6 REM # of the different sounds the Apple /// #
7 REM # can make. Just type it in and enter #
8 REM # the command 'RUN'. #
9 REM #####
100 PRINT CHR$(14);:REM Turn off the screen
110 count%=0:time%=1:c.val%=16383:mode%=128:vol%=63
120 OPEN#1,".audio":GOTO 499
200 PRINT#1;CHR$(mode%);CHR$(vol%);
210 PRINT#1;CHR$(count%-256*INT(count%/256));
220 PRINT#1;CHR$(INT(count%/256));
230 PRINT#1;CHR$(time%-256*INT(time%/256));
240 PRINT#1;CHR$(INT(time%/256));
250 RETURN
499 ON KBD GOTO 1000
500 FOR a%=1 TO 1000
510 count%=INT(RND(1)*c.val%)
520 GOSUB 200
530 NEXT
540 GOTO 500
1000 IF KBD=27 THEN CLOSE#1:TEXT:HOME:END
1010 ON KBD GOTO 1000
1020 RETURN
    
```